Symbolic computation of asymptotic formula for large solutions of the p-Laplacian

Petr Girg

Abstract. Explicit formulae for solutions of quasilinear boundary-value problems can be found very seldom and qualitative theory shall be used instead. Topological methods based on bifurcations from infinity are powerfull tool in the study of asymptoticaly homogeneous boundary value problems. The key point to this methods are asymptotic estimates of large solutions. In this paper we present asymptotic formula for large solutions of the p-Laplacian.

1 Introduction

Existence, multiplicity, and bifurcation of solutions to the boundary value problem (BVP for short):

$$\begin{aligned} -\left(|u'(x)|^{p-2}u'(x)\right)' &= \lambda |u(x)|^{p-2}u(x) + f(x)\,, && \text{in } (0,\pi_p)\,, \\ u(0) &= u(\pi_p) = 0\,, && \end{aligned} \tag{1}$$

are questions interesting from both theoretical and practical point of view. Here $p > 1$, $\pi_p = 2\int_0^1 \frac{ds}{(1-s^p)^{1/p}} = \frac{2\pi}{p\sin\frac{\pi}{p}}$ ($\pi_2 = \pi = 3.1415926\ldots$), λ is a real number (bifurcation parameter) and $f \in L^\infty(0,\pi_p)$. As both terms containing u' and u are $(p-1)$-homogeneous with respect to real multiples of u, the problem (4) belongs to class of asymptoticaly homogeneous problems cf [10]. Topological methods based on bifurcations from infinity are powerfull tool in the study of asymptoticaly homogeneous boundary value problems [1, 2, 4, 8, 9, 12, 16]. The standard compacteness argument (see e.g. [16]) shows that the prospective asymptotic-bifurcation points are the eigenvalues μ_k of the nonlinear eigenvalue problem

$$\begin{aligned} (|u'|^{p-2}u')' + \lambda|u|^{p-2}u &= 0\,, && \text{in } (0,\pi_p)\,, \\ u(0) = u(\pi_p) &= 0\,. && \end{aligned} \tag{2}$$

Note that the set of all eigenvalues of (2) is fully described by

$$\mu_k = (p-1)k^p \qquad k \in \mathbb{N}\,;$$

each eigenvalue being simple with corresponding eigenfunction $\sin_p k(\cdot)$ where $\sin_p(\cdot)$ is defined as the solution to the initial value problem

$$\begin{aligned} (|u'|^{p-2}u')' + (p-1)|u|^{p-2}u &= 0\,, && \text{in } (0,\pi_p)\,, \\ u(0) = 0, \quad u'(0) &= 1\,. && \end{aligned} \tag{3}$$

Some asymptotic estimates of large solutions to

$$\begin{aligned} -\left(|u'(x)|^{p-2}u'(x)\right)' &= (p-1)k^p|u(x)|^{p-2}u(x) + f(x)\,, && \text{in } (0,\pi_p)\,, \\ u(0) &= u(\pi_p) = 0\,, && \end{aligned} \tag{4}$$

are necessary for application of the abstract bifurcation theory to (4). In the forthcomming calculations we will shortcut $\sin_p$ to s_p in order to keep expressions reasonable long. We define $\mathrm{c}_p \stackrel{\text{def}}{=} \mathrm{s}_p'$ and use the so called p-trigonometric identity $|\mathrm{s}_p(x)|^p + |\mathrm{c}_p(x)|^p = 1$, for details see [5] (note that there is an alternative definition of c_p by LINDQVIST [13] which satisfy other p-trigonometric identity [13]). From the p-trigonometric identity, we infer that $\mathrm{c}_p'(x) = -\frac{|\mathrm{s}_p(x)|^{p-2}}{|\mathrm{c}_p(x)|^{p-2}}\mathrm{s}_p(x)$ (for $p = 2$ we have $\cos' x = c_2'(x) = -s_2(x) = -\sin x$ as expect). Note that functions s_p and c_p satisfy many more identities. In our calculations we use the above two mentioned identities together with some information on integrability of s_p and c_p and their respective derivatives (see section 4).

In [14] asymptotic formula of second order of large solutions to initial-value problem

$$\begin{aligned} -\left(|u'(x)|^{p-2}u'(x)\right)' = (p-1)k^p|u(x)|^{p-2}u(x) + f(x) \qquad &\text{in } (0,T)\,,\ T > \pi_p\,, \\ u(0) = 0\,, u'(0) = \alpha\,, \end{aligned} \tag{5}$$

as $\alpha \to +\infty$ or $\alpha \to -\infty$ has been provided together with thorough discussion of its consequences to (4). In order to answer several theoreticaly important open questions from [14] one needs asymptotic formulae of higher order. To establish such formulae, one shall perform very long and tedious calculations. Avoiding errors due to one's negligence is almost impossible. This paper concerns with symbolic computation of asymptotic formula of the fourth order in *Mathematica*®; see also [11]. The theoretical background of this asymptotics and thorough discussion of its consequences to existence, multiplicity, and bifurcation of solutions from infinity, are answered in the forthcomming paper BENEDIKT-GIRG-TAKÁČ [3].

2 The generalized Prüfer transfomation

Following [14], the large solutions of

$$\begin{aligned} -\left(|u'(x)|^{p-2}u'(x)\right)' = (p-1)k^p|u(x)|^{p-2}u(x) + f(x) \qquad &\text{in } (0,T)\,,\ T > \pi_p\,, \\ u(0) = 0\,, u'(0) = \alpha\,, \end{aligned}$$

can be studied by means of the so called generalized Prüfer transformation. We assume that the solutions can be written in the form

$$u(x) = \frac{1}{k} r(x)^{\frac{1}{p-1}} \mathrm{s}_p(k(x + \vartheta(x))) \tag{6}$$

$$u'(x) = r(x)^{\frac{1}{p-1}} \mathrm{c}_p(k(x + \vartheta(x))) \tag{7}$$

where $r \in C^1(\mathbb{R}^+, \mathbb{R}^+)$, $\vartheta \in C^1(\mathbb{R}^+, \mathbb{R})$ (for detailed theoretical justification see [3]).

At first we shall find corresponding equations for r and ϑ. A convenient way for this, is to define $\varphi_p(z) \stackrel{\text{def}}{=} |z|^{p-2}z$ for $z \neq 0$, $\varphi_p(0) \stackrel{\text{def}}{=} 0$ and to observe that $\varphi_p(x \cdot y) = \varphi_p(x) \cdot \varphi_p(y)$, $x\varphi_p(x) = |x|^p$, $\varphi_p'(x) = (p-2)|x|^{p-2}$. Thus expressions containing φ_p will be simplified by `psimpl[expr_]:=expr //. rulesp` using following rules:

```
rulesp = {
          HoldPattern[Phi[p][a_ * b_ ]] :> Phi[p][a] * Phi[p][b],
          HoldPattern[a_ * Phi[p_][a_]] :> Abs[a]^p,
          HoldPattern[a_ * Phi[p_]'[a_]] :> (p - 1) Phi[p][a],
          HoldPattern[Phi[p_]'[a_]] :> (p - 1) Abs[a]^(p - 2),
          HoldPattern[Phi[p_][k_?(Refine[# > 0] &)]] :> k^(p - 1)
         }
```

Since obviously $u'(x) = u'(x)$, the derivative of the right-hand side of (6)

```
In[1]:=   subu[x_] = 1/k r[x]^(1/(p - 1)) s[p][k (x + Theta[x])]
```

equals to the right-hand side of (7):

```
In[2]:=   subDu[x_] = r[x]^(1/(p - 1))c[p][k(x + Theta[x])]
```

i.e.

```
In[3]:=   D[subu[x], x] == subDu[x]
              // Simplify[#, {k > 0, p > 1, r[x] > 0}] &
```

which yields

$$\texttt{Out[3]:=}\ r(x)^{\frac{1}{p-1}}\left(\mathrm{s}_p(k(x+\vartheta(x)))r'(x)+k(p-1)r(x)\mathrm{c}_p(k(x+\vartheta(x)))\vartheta'(x)\right)=0\,. \quad (8)$$

This equation can be solved for $r(x)$ and we will keep its solution in the substitution rule

```
In[4]:=   rsubst = (Solve[%, r[x]] // Flatten)
```

Now we substitute (6) and (7) for u and u' in the equation from (2). Convenient way for this is to rewrite the equation (7) in the form

$$-\left(|u'(x)|^{p-2}u'(x)\right)' - (p-1)k^p|u(x)|^{p-2}u(x) - f(x) = 0$$

and work with its left-hand side only, this we denote by `rce`. The computation follows.

```
In[5]:=
rce = Hold[
          -D[Phi[p][u'[x]], x] - k^p (p - 1) Phi[p][u[x]] - f[x]
         ]
(rce /. {u'[x] -> subDu[x],  u[x] -> subu[x]} // ReleaseHold )
(*it substitutes into held equation, then releases*)
/. c[p]'[x_] :>  -(Abs[s[p][x]]/Abs[c[p][x]])^(p - 2)*s[p][x]
(*express the first derivative of c[p] in terms of s[p] and c[p]*)
```

Out[5]:=

$$-(p-1)\varphi_p\left(\frac{r(x)^{\frac{1}{p-1}}\mathrm{s}_p(k(x+\vartheta(x)))}{k}\right)k^p - f(x) - \left(\frac{r(x)^{\frac{1}{p-1}-1}\mathrm{c}_p(k(x+\vartheta(x)))r'(x)}{p-1} - k\left(\frac{1}{|\mathrm{c}_p(k(x+\vartheta(x)))|}\right)^{p-2}\right.$$

$$\left.|\mathrm{s}_p(k(x+\vartheta(x)))|^{p-2}r(x)^{\frac{1}{p-1}}\mathrm{s}_p(k(x+\vartheta(x)))\left(\vartheta'(x)+1\right)\right)\varphi_p'\left(r(x)^{\frac{1}{p-1}}\mathrm{c}_p(k(x+\vartheta(x)))\right)$$

```
In[6]:=
(
(Assuming[{r[x] > 0, p > 1, k > 0}, Simplify /@ psimpl[%] ])
/.
r[x] ->
       -(s[p][k*(x + Theta[x])]*r'[x])/
        (k*(-1 + p)*c[p][k*(x + Theta[x])]*Theta'[x])
)
//
(Simplify[#, {p > 1, k > 0,
   Element[{c[p][x_], s[p][x_]},Reals}] & /@
        (Expand[#*c[p][k(x + Theta[x])]] // psimpl) &
)
```

Out[6]:= $-r'(x)|\mathrm{c}_p(k(x+\vartheta(x)))|^p - f(x)\,\mathrm{c}_p(k(x+\vartheta(x))) - |\mathrm{s}_p(k(x+\vartheta(x)))|^p r'(x)$

By the p-trigonometric identity $|\mathrm{s}_p(x)|^p + |\mathrm{c}_p(x)|^p = 1$,

```
In[7]:=   % /. Abs[s[p][x_]]^p a_. + Abs[c[p][x_]]^p a_. :> a
```

we find

Out[7]:= $-f(x)\,\mathrm{c}_p(k(x+\vartheta(x))) - r'(x)\,.$

Thus we have

$$r'(x) = -f(x)\,\mathrm{c}_p(k(x+\vartheta(x)))\,. \tag{9}$$

Now plugging this into (8), we find

$$\vartheta'(x) = \frac{1}{k(p-1)}\frac{f(x)\,\mathrm{s}_p(k(x+\vartheta(x)))}{r(x)}\,. \tag{10}$$

The system (9)–(10) with $r(0) = r_0 = \alpha^{p-1}$ and $\vartheta(0) = 0$ is in some sense equivalent to (2) (for precise formulation of this statement see [3]). Now we will find asymptotic estimates for solutions of (9)–(10) with $r(0) = r_0 = \alpha^{p-1}$ and $\vartheta(0) = 0$ with $r_0 = \alpha^{p-1} \to +\infty$.

3 First order asymptotics

The first order asymptotics can easily be done by hand. Since it gives a good idea where the asymptotic formulae are from, we present the calculations here. Integrating (9) and (10) from 0 to x we find

$$r(x) = r_0 - \int_0^x f(s)\,\mathrm{c}_p(k(s+\vartheta(s)))\,\mathrm{d}\,s \qquad \text{and} \tag{11}$$

$$\vartheta(x) = \frac{1}{k(p-1)}\int_0^x \frac{f(s)\,\mathrm{s}_p(k(s+\vartheta(s)))}{r(s)}\,\mathrm{d}\,s\,. \tag{12}$$

Since f is integrable and c_p is bounded we find that there exists $K > 0$ such that $|r(x) - r_0| < K$. The Taylor formula for $1/z$ in the neighbourhood of z_0 reads

$$\frac{1}{z} = \frac{1}{z_0} - \frac{(z-z_0)}{(z_0+\omega(z-z_0))^2} \qquad \text{for some } \omega \in (0,1)\,.$$

From this formula and from the fact that $|r(x) - r_0| < K$, we infer that

$$\frac{1}{r(x)} = \frac{1}{r_0} - \frac{r(x) - r_0}{r_0 + \omega(r(x) - r_0)^2} = \frac{1}{r_0} + O(1/r_0^2)\,.$$

Thus from (12) we find

$$\vartheta(x) = \frac{1}{k(p-1)r_0} \int_0^x f(t)\mathrm{s}_p(k(t + \vartheta(t)))(1 - O(1/r_0))\,\mathrm{d}\,t\,.$$

Therefore, by integrabilty of f and boundedness of s_p, we find

$$|\vartheta(x)| \le \frac{L}{r_0} \tag{13}$$

for some $L > 0$. Let us refine an estimate of $r(x)$ now. Using the Taylor formula

$$\mathrm{c}_p(z) = \mathrm{c}_p(z_0) + \mathrm{c}_p'(z_0 + \omega(z - z_0))(z - z_0)\,,$$

with $\omega \in (0,1)$, $z = k(x + \vartheta(x))$ and $z_0 = k(x + \vartheta(0)) = kx$, and using facts that, for $1 < p < 2$, c_p' is continuous (and thus bounded), and that ϑ is estimated by (13), we find

$$r(x) = r_0 - \int_0^x f(t)\left\{\mathrm{c}_p(kt) + \mathrm{c}_p'(k(t + \omega_t\,\vartheta(t)))\right\}\mathrm{d}\,t = r_0 - \int_0^x f(t)\mathrm{c}_p(kt)\,\mathrm{d}\,t + O(1/r_0)\,. \tag{14}$$

Note that $\omega_t \in (0,1)$ for each $t \in (0,x)$ and that ω_t can possibly be non-unique. Now using the Taylor formula $1/z = 1/z_0 + (z - z_0)/z_0^2 + (z - z_0)^3/(z_0 + \gamma(z - z_0))^3$ with some $\gamma \in (0,1)$, $z = r(x)$, $z_0 = r_0$ and the fact that $|r(x) - r_0| < K$, we infer from (14) that

$$\frac{1}{r(x)} = \frac{1}{r_0} - \frac{\int_0^x f(t)\mathrm{c}_p(kt)\,\mathrm{d}\,t}{r_0^2} + O(1/r_0^3)\,. \tag{15}$$

With this at hand we can refine estimate for ϑ. By the Taylor formula we have

$$\mathrm{s}_p(kz) = \mathrm{s}_p(kz_0) + k\,\mathrm{c}_p(kz_0 + \omega(z - z_0))(z - z_0) \qquad \text{for some } \omega \in (0,1)\,.$$

Thus

$$\vartheta(x) = \frac{1}{k(p-1)} \int_0^x \left(f(t)\mathrm{s}_p(kt) + kf(x)\vartheta(t)\mathrm{c}_p(k(t + \omega_t\vartheta(t)))\right) \times$$
$$\left(\frac{1}{r_0} - \frac{\int_0^t f(\tau)\mathrm{c}_p(k\tau)\,\mathrm{d}\,\tau}{r_0^2} + O(1/r_0^3)\right)\mathrm{d}\,t$$

which is

$$\vartheta(x) = \frac{1}{k(p-1)r_0} \int_0^x f(t)\mathrm{s}_p(k\,t)\,\mathrm{d}\,t + O(1/r_0^2)\,. \tag{16}$$

In this way we can continue and get asymptotic formula of the second order as in [14]. For the third order formula the things are getting much more complicated by fast grow of the number of terms. Finaly, the calculation of the fourth order formula is realy long (consists of 208 A4 pages printed with 10pt font) and is probably unfeasible without assistance of computer. For this reason, we shall limit ourselves to sketch main steps here and to comment some implementation notes in next section. At the end of this paper, we will present the fourth order formula.

Our algorithm:

Step 1: The second order asymptotics for ϑ. We substitute second order Taylor formula $\mathrm{s}_p(z) = \mathrm{s}_p(z_0) + (z - z_0)\mathrm{c}_p(z_0) + 1/2(z - z_0)^2\mathrm{c}_p'(z - \omega(z - z_0))$ in place of s_p in (12). Note that $z = k(x + \vartheta(x))$ and $z_0 = kx$. Then we replace every occurence of $\vartheta(x)$ by (16) and every occurence of $1/r(x)$ by (15).

Step 2: The second order asymptotics for r. We substitute second order Taylor formula $\mathrm{c}_p(z) = \mathrm{c}_p(z_0) + (z - z_0)\mathrm{c}_p'(z_0) + 1/2(z - z_0)^2\mathrm{c}_p''(z - \omega(z - z_0))$ in place of c_p in (11). Then we replace every occurence of $\vartheta(x)$ by its second order estimate from the previous step and every occurence of $1/r(x)$ by (15).

Step 3: Third order asymptotics for $1/r$. We substitute fourth order Taylor formula $\frac{1}{z} = \frac{1}{z_0} - \frac{z-z_0}{z^2} + \frac{(z-z_0)^2}{z^3} - \frac{(z-z_0)^3}{(z-\omega(z-z_0)^4)}$ in place of $1/r(x)$ with $z = r(x)$ and $z_0 = r_0$. Then we replace every occurence of $(r(x) - r_0)^3$ by $O(1)$, every occurence of $(r(x) - r_0)^2$ by using (15) and every occurence of $r(x) - r_0$ by using second order estimate or $r(x)$ from previous step.

Step 4: Third order asymptotics for ϑ. We substitute third order Taylor formula $\mathrm{s}_p(z) = \mathrm{s}_p(z_0) + (z - z_0)\mathrm{c}_p(z_0) + 1/2(z - z_0)\mathrm{c}_p'(z_0) + 1/6(z - z_0)^3\mathrm{c}_p''(z - \omega(z - z_0))$ in place of s_p in (12). Note that $z = k(x + \vartheta(x))$ and $z_0 = kx$. Then we replace $1/r(x)$ by its third order asymptotics and expand the multiplication. In the resulting expression we get products of powers of r_0 and $\vartheta(x)$. Now we replace every occurence of the n-th power of $\vartheta(x)$ by one of the following $O(1/r_0^n)$, (16) or second order asymptotics of ϑ in such way that we get asymptotics of third order, *i.e.*, an expression of the form $\vartheta(x) = a(x)/r_0 + b(x)/r_0^2 + c(x)/r_0^3 + O(1/r_0^4)$ with $a(x), b(x), c(x)$ given explicitely in terms of $f, \mathrm{s}_p, \mathrm{c}_p$.

Step 5: Third order asymptotics for r. Analogous to Step 2.

Step 6: Fourth asymptotics for $1/r$. Analogous to Step 3.

Step 7: Fourth order asymptotics for ϑ. Analogous to Step 4.

4 Implementation note to higher order asymptotics

In deriving asymptotic formula for ϑ, we need to work with symbols O and o. As *Mathematica*® has built in plenty of functions it is rather temptating to write `O[x]-O[x]` and hope to get mathematicaly correct answer `O[x]`, lets do it:

```
In[1]:=     O[x]-O[x]
Out[1]:=    O[x]
```

The result seems well. However, looking at the full form of `O[x]`:

```
In[2]:=     FullForm[O[x]]
Out[2]:=    SeriesData[x, 0, List[], 1, 1, 1]
```

we see that the head of `O[x]` is not a symbol `O`, it is just `OutputForm` of expression `SeriesData[x, 0, List[], 1, 1, 1]`. Since for example `myO[x]+myO[x^2]` evaluates to nested `SeriesData` object:

```
In[3]:=   myO[x]+myO[x^2] \\ FullForm
Out[3]:=  SeriesData[x, 0, List[
            SeriesData[Power[x, 2], 0, List[], 1, 1, 1]], 0, 1, 1]
```

it would be very difficult to manipulate large expressions with many nested `SeriesData` objects in this way. For that reason we have decided to implement mathematical symbols O and o by ourselves, lets call them `myO[x]` and `myo[x]`.

In order to prevent `myO[x]-myO[x]` to be evaluated to `0` the following rule for `myO` has to be defined such as to be applied automatically in the main *Mathematica*® evaluation loop:

```
In[4]:= myO /: HoldPattern[myO[x_] - myO[x_]] := myO[x]
```

As we want `myO[myO[x]]` to evaluate to `myO[x]` we set its attributes to `Flat`. Furthermore, $a(x)O(x) = O(x)$ provided $a(x)$ is bounded, thus we need to implement the concept of boundedness. As *Mathematica*® works as a knowledge database through rules, we can store (and/or deduce) information about boundedness of mathematical objects in the following way:

```
In[5]:=
BoundedQ[a_?NumberQ]=True ;

BoundedQ[Times[a_, b_]] := BoundedQ[a] && BoundedQ[b]

BoundedQ[Plus[a_, b_]] := BoundedQ[a] && BoundedQ[b]

BoundedQ[Integrate[f_, {x_, a_, b_}]] :=
         BoundedQ[f] && BoundedQ[a] && BoundedQ[b]
```

Boundedness of a particular objects, e.g., constants k, p and functions $\mathrm{s}_p(x), f(x)$ can be incorporated in a natural way:

```
In[6]:=
BoundedQ[k]=True ;

BoundedQ[p]=True;

BoundedQ[sp[x_]]=True;

BoundedQ[f[x_]]=True;
```

We provide some examples of evaluation of `BoundedQ`. Obviously, $(k+1)(p-3)$ is bounded provided k and p are:

```
In[7]:= BoundedQ[(k+1)(p-3)]
Out[7]:= True
```

Also $\int_0^1 f(x)\mathrm{s}_p(x)\,\mathrm{d}\,x$ is bounded provided $f(x)$ and $\mathrm{s}_p(x)$ are bounded functions:

```
In[8]:=BoundedQ[Integrate[f[x] sp[x], {x, 0, 1}]]
Out[8]:= True
```

We would also like to take advatage of the fact that $\mathrm{c}_p''(x)$ is integrable function (but not bounded), $\mathrm{c}_p'' \in L^1(a,b)$ for $a,b \in \mathbb{R} : a < b$. Then we have that $\int_a^b f(x)\mathrm{c}_p''(x)\,\mathrm{d}\,x$ is bounded provided a, b are bounded and f is a bounded function:

```
In[9]:=BoundedQ[Integrate[f_ cp''[x_], {x_, a_, b_}]] :=
         BoundedQ[f] && BoundedQ[a] && BoundedQ[b]
```

In our symbolic calculation we have to deal with expressions of the following type:

$$\int_0^1 \left(\int_0^x f(s)\,\mathrm{c}_p''(s)\,\mathrm{d}\,s \right) \mathrm{d}\,x$$

```
In[10]:=BoundedQ[Integrate[Integrate[f[s] cp''[s], {s, 0, x}], {x, 0, 1}]]
Out[10]:=BoundedQ[x]
```

The problem consists in the fact that the information that $x \in [0,1]$ from the outer integral is not used. This problem can be solved by the use of boundedness of objects related to some variable and its domain. But this general approach significantly slows down the calculations. For that reason, we use the following cheap trick in our calculations:

```
In[11]:=Module[{x}, BoundedQ[x] = True;
         BoundedQ[Integrate[Integrate[f[s] cp''[s], {s, 0, x}], {x, 0, 1}]]
             ]
Out[11]:=True
```

Now we are ready to define remaining rules for `myO` to get expected behaviour:

```
rulesForMyO={

HoldPattern[HoldPattern[myO[(x_)^(n_.)]*(y_)]] :> myO[x^n] /; BoundedQ[y],

HoldPattern[HoldPattern[myO[x_]*(x_)^(m_.)]] :> myO[x^(m + 1)],

HoldPattern[HoldPattern[myO[(x_)^(n_.)]*(x_)^(m_.)]] :> myO[x^(n + m)],

HoldPattern[HoldPattern[myO[x_]*myO[(x_)^(n_.)]]] :> myO[x^(n + 1)],
```

```
HoldPattern[HoldPattern[myO[(x_)^(m_)]*myO[(x_)^(n_.)]]] :> myO[x^(n + m)],

HoldPattern[HoldPattern[myO[x_]^(m_.)*myO[(x_)^(n_)]]] :> myO[x^(n + m)],

HoldPattern[HoldPattern[myO[x_]^(m_)]] :> myO[x^m],

HoldPattern[HoldPattern[myO[(x_)^(m_.)] - myO[(x_)^(m_.)]]] :> myO[x^m],

HoldPattern[HoldPattern[myO[(x_)^(m_.)] + myO[(x_)^(n_.)]]] :>
  myO[x^Max[{m, n}]],

HoldPattern[HoldPattern[Integrate[myO[y_], {x_, a_, b_}]]] :>
myO[y] /; FreeQ[y, x],

HoldPattern[HoldPattern[Integrate[myO[y_], {x_, a_, b_}]]] :>
  myO[y] /; BoundedQ[a] && BoundedQ[b],

HoldPattern[HoldPattern[Limit[myO[1], x_ -> Infinity]]] :> myO[1],

HoldPattern[HoldPattern[Limit[myO[(x_)^(n_.)], x_ -> Infinity]]] :>
  myO[1]*Limit[x^n, x -> Infinity],

HoldPattern[Infinity*myO[1]] :> Infinity }
```

With this rules at hand, simplifier for `myO` reads as follows:

```
sMyO[expression_]:=expression //. rulesForMyO
```

5 The formula

Denoting

$$\begin{aligned}
L(s,f) &= \int_0^s f(z)\,\mathrm{s}_p(kz)\,dz\,;\\
M(s,f) &= \int_0^s f(z)\,\mathrm{c}_p(kz)\,dz\,;\\
H(s,f) &= \int_0^s f(z)\,\mathrm{c}_p'(kz)\,dz\,;\\
Q(s,f,f) &= \int_0^s f(\sigma)\left(\int_0^\sigma f(z)\,\mathrm{s}_p(kz)\,dz\right)\mathrm{c}_p(k\sigma)\,d\sigma\,,
\end{aligned}$$

we can write our formula as follows

$$\vartheta(s) = \frac{1}{k(p-1)r_0} L(s,f) +$$

$$\frac{1}{k(p-1)^2 r_0^2} \left\{ (p-1)L(s,f)M(s,f) - (p-2)Q(s,f,f) \right\} +$$

$$\frac{1}{2k(p-1)^3 r_0^3} \left\{ \int_0^s L(x,f)^2 H'(x,f) \,\mathrm{d}\, x + \right.$$

$$2(p-1) \left(\int_0^s \left(\int_0^x L(s,f) H'(s,f) \,\mathrm{d}\, s \right) L'(x,f) \,\mathrm{d}\, x + \right.$$

$$\left. (p-1) \int_0^s M(x,f)^2 L'(x,f) \,\mathrm{d}\, x + 2 \int_0^s L(x,f) M(x,f) M'(x,f) \,\mathrm{d}\, x \right) -$$

$$\left. 2(p-2) \int_0^s Q(x,f,f) M'(x,f) \,\mathrm{d}\, x \right\} +$$

$$\frac{1}{6k(p-1)^4 r_0^4} \left\{ 3 \left(3(p-1) \int_0^s L(x,f)^2 M(x,f) H'(x,f) \,\mathrm{d}\, x \right. \right.$$

$$-2(p-2) \int_0^s L(x,f) Q(x,f,f) H'(x,f) \,\mathrm{d}\, x$$

$$+2 \int_0^s \left(\int_0^x L(s,f) M(s,f) H'(s,f) \,\mathrm{d}\, s \right) L'(x,f) \,\mathrm{d}\, x -$$

$$4 \int_0^s \left(\int_0^x Q(s,f,f) H'(s,f) \,\mathrm{d}\, s \right) L'(x,f) \,\mathrm{d}\, x -$$

$$\int_0^s \left(\int_0^x L(s,f)^2 U'(s,f) \,\mathrm{d}\, s \right) L'(x,f) \,\mathrm{d}\, x +$$

$$4 \int_0^s \left(\int_0^x L(s,f) H'(s,f) \,\mathrm{d}\, s \right) M(x,f) L'(x,f) \,\mathrm{d}\, x -$$

$$2 \int_0^s M(x,f)^3 L'(x,f) \,\mathrm{d}\, x +$$

$$\int_0^s \left(\int_0^x L(\sigma,f)^2 H'(\sigma,f) \,\mathrm{d}\, \sigma \right) M'(x,f) \,\mathrm{d}\, x -$$

$$2\left(\int_0^s \left(\int_0^x \left(\int_0^\sigma L(\tau,f)H'(\tau,f)\,\mathrm{d}\,\tau\right) L'(\sigma,f)\,\mathrm{d}\,\sigma\right) M'(x,f)\,\mathrm{d}\,x-\right.$$

$$\int_0^s \left(\int_0^x M(\sigma,f)^2 L'(\sigma,f)\,\mathrm{d}\,\sigma\right) M'(x,f)\,\mathrm{d}\,x+$$

$$2\int_0^s \left(\int_0^x L(\sigma,f)M(\sigma,f)M'(\sigma,f)\,\mathrm{d}\,\sigma\right) M'(x,f)\,\mathrm{d}\,x-$$

$$2\int_0^s \left(\int_0^x Q(\sigma,f,f)M'(\sigma,f)\,\mathrm{d}\,\sigma\right) M'(x,f)\,\mathrm{d}\,x+$$

$$\int_0^s \left(\int_0^x L(s,f)H'(s,f)\,\mathrm{d}\,s\right) L(x,f)M'(x,f)\,\mathrm{d}\,x-$$

$$2\int_0^s L(x,f)M(x,f)^2 M'(x,f)\,\mathrm{d}\,x+$$

$$\left. 2\int_0^s M(x,f)Q(x,f,f)M'(x,f)\,\mathrm{d}\,x\right)+$$

$$p\left(2(p-2)\int_0^s \left(\int_0^x L(s,f)M(s,f)H'(s,f)\,\mathrm{d}\,s\right) L'(x,f)\,\mathrm{d}\,x-\right.$$

$$2(p-3)\int_0^s \left(\int_0^x Q(s,f,f)H'(s,f)\,\mathrm{d}\,s\right) L'(x,f)\,\mathrm{d}\,x+$$

$$\int_0^s \left(\int_0^x L(s,f)^2 U'(s,f)\,\mathrm{d}\,s\right) L'(x,f)\,\mathrm{d}\,x+$$

$$2\left(2(p-2)\int_0^s \left(\int_0^x L(s,f)H'(s,f)\,\mathrm{d}\,s\right) M(x,f)L'(x,f)\,\mathrm{d}\,x+\right.$$

$$((p-3)p+3)\int_0^s M(x,f)^3 L'(x,f)\,\mathrm{d}\,x+$$

$$\int_0^s \left(\int_0^x \left(\int_0^\sigma L(\tau,f)H'(\tau,f)\,\mathrm{d}\,\tau\right) L'(\sigma,f)\,\mathrm{d}\,\sigma\right) M'(x,f)\,\mathrm{d}\,x+$$

$$p\int_0^s \left(\int_0^x M(\sigma,f)^2 L'(\sigma,f)\,\mathrm{d}\,\sigma\right) M'(x,f)\,\mathrm{d}\,x-$$

$$2\int_0^s \left(\int_0^x M(\sigma,f)^2 L'(\sigma,f)\,\mathrm{d}\,\sigma\right) M'(x,f)\,\mathrm{d}\,x+$$

$$2\int_0^s \left(\int_0^x L(\sigma,f)M(\sigma,f)M'(\sigma,f)\,\mathrm{d}\,\sigma\right) M'(x,f)\,\mathrm{d}\,x-$$

$$\int_0^s \left(\int_0^x Q(\sigma,f,f)M'(\sigma,f)\,\mathrm{d}\,\sigma\right) M'(x,f)\,\mathrm{d}\,x+$$

$$\int_0^s \left(\int_0^x L(s,f)H'(s,f)\,\mathrm{d}\,s\right) L(x,f)M'(x,f)\,\mathrm{d}\,x+$$

$$2p\int_0^s L(x,f)M(x,f)^2 M'(x,f)\,\mathrm{d}\,x-$$

$$4\int_0^s L(x,f)M(x,f)^2 M'(x,f)\,\mathrm{d}\,x-$$

$$\left.\left.\left.(p-3)\int_0^s M(x,f)Q(x,f,f)M'(x,f)\,\mathrm{d}\,x\right)\right)\right)+$$

$$\left.\int_0^s L(x,f)^3 U'(x,f)\,\mathrm{d}\,x\right\} + o\left(\frac{1}{r_0^4}\right)$$

This asymptotic formula holds provided $r_0 \to \pm\infty$, $1 < p < 3/2$ and f is a smooth function vanishing in the neighbourhood of roots of $\mathrm{s}_p(k\,x) = 0$. If we consider the third order asymptotics (terms with $1/r_0, 1/r_0^2, 1/r_0^3$) then the formula is valid for $r_0 \to \pm\infty$, $1 < p < 2$ and smooth functions f.

6 Conclusion

We found fourth order asymptotic formula for large solutios of (4). This formula helps to answer several theoretical questions from [14]. Forthcomming paper by Benedikt, Girg and Takáč [3] answers this theoretical questions. If all steps printed in 10pt, they cover 208 pages of A4. Thus it seems that such formula would hardly be calculated by hand without any assistance of computer.

7 Acknowledgement

The author would like to thank to the Humboldt Foundation which sponsored his research on this topic at University of Rostock from July 1, 2003, to June 30, 2004. Although this formula was found recently, the research already started in Rostock. The extensive calculations were made possible mainly thanks to computational resources supported from the Grant # 1N04078 of the Ministry of Education, Youth and Sports of the Czech Republic.

Reference

[1] AMBROSETTI, A., ARCOYA, D., On a Quasilinear Problem at Strong Resonance, *Topol. Methods in Nolin. Anal.*, **6** (1995), pp. 255–264.

[2] ARCOYA, D., GÁMEZ, J.-L. Bifurcation Theory and Related Problems: Anti-Maximum Principle and Resonance, *Comm. Partial Diff. Equations*, **26**(9–10) (2001), pp. 1879–1911.

[3] BENEDIKT J., GIRG, P., TAKÁČ, P., *On the Fredholm alternative at higher eigenvalues for the p-Laplacian in one dimension*, In Preparation.

[4] ČEPIČKA, J., DRÁBEK, P., GIRG, P., Open problems related to the p-Laplacian, Bol. Soc. Esp.Mat. Apl. **29** (2004), 13–34

[5] M. A. del Pino, P. Drábek, and R. F. Manásevich, The Fredholm alternative at the first eigenvalue for the one-dimensional p-Laplacian, *J. Differential Equations*, **151** (1999), 386–419.

[6] del PINO, M. A., ELGUETA, M., MANÁSEVICH, R. F., A homotopic deformation along p of a Leray-Schauder degree result and existence for $(|u'|^{p-2}u')' + f(t,u) = 0$, $u(0) = u(T) = 0$, $p > 1$, *J. Differential Equations*, **80**(1) (1989), 1–13.

[7] del PINO, M. A., MANÁSEVICH, R. F., Global bifurcation from the eigenvalues of the p-Laplacian, *J. Differential Equations*, **92** (1991), 226–251.

[8] DRÁBEK, P., GIRG, P., TAKÁČ, P., Bounded perturbations of homogeneous quasilinear operators using bifurcations from infinity, *J. Differential Equations*, **204** (2004), 265–291.

[9] DRÁBEK, P., GIRG, P., TAKÁČ, P., ULM, M., The Fredholm alternative for the p-Laplacian: bifurcation from infinity, existence and multiplicity of solutions, *Indiana Univ. Math. J.*, **53**(2) (2004), 433–482.

[10] FUČÍK, S., NEČAS, J., SOUČEK, J., SOUČEK, V., *Spectral Analysis of Nonlinear Operators*, Lecture Notes in Mathematics 346, Springer-Verlag, Berlin-Heidelberg-New York, 1973.

[11] GIRG, P., Symbolic computation of asymptotic formula for large solutions of the p-Laplacian using *Mathematica*®, *In* Aplimat 2006 (Monika Kováčová ed.), Bratislava : Slovak University of Technology , 2006. pp. 237–248. ISBN 80-967305-5-X.

[12] GÁMEZ, J.-L., GIRG, P., *Fredholm Alternative for the p-Laplacian at the first eigenvalue and bifurcations from the infinity*, preprint, University of West Bohemia, 2000.

[13] LINDQVIST, P., Some remarkable sine and cosine functions, *Ricerche di Matematica* 44 (1995), 269 – 290.

[14] MANÁSEVICH, R. F., TAKÁČ, P., On the Fredholm alternative for the p-Laplacian in one dimension, *Proc. London Math. Soc.*, **84** (2002), 324–342.

[15] MAEDER, R., *Programming in Mathematica*, 3rd ed., Addison-Wesley Publishing Company, Reading, 1997

[16] LE, V. K., SCHMITT, K., *Global Bifurcation in Variational Inequalities*, Applied Mathematical Series 123, Springer Verlag, New York, 1997.

[17] WOLFRAM S., *The Mathematica Book*, the fifth edition, Wolfram Media, Champaign, 2003

Petr Girg
KMA-FAV
Západočeská Univerzita v Plzni
Univerzitní 22
30614 Plzeň
CZECH REPUBLIC

e-mail: pgirg@kma.zcu.cz
tel.: ++420 377 632 644